// File: /contracts/Main.sol



// SPDX-License-Identifier: MIT LICENSE

pragma solidity ^0.8.17;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "./interfaces/iNft.sol";

contract Main is Ownable, Pausable, ReentrancyGuard {

    constructor() {
        _pause();
    }

    /** CONTRACTS */
    iNft public nftContract;

    /** EVENTS */
    event ManyNftsMinted(address indexed owner, uint16[] tokenIds);

    /** PUBLIC VARS */
    uint256 public mintPriceType1 = 0.00001 ether;
    uint256 public mintPriceType2 = 0.1 ether;
    uint256 public totalType1Minted = 4;
    uint256 public totalType2Minted = 1;

    bool public PRE_SALE_STARTED;
    // maximum nfts at pre-sale - 81 are for the team + marketing and 300 for pre-sale wallets
    uint256 public MAX_PRE_SALE_MINTS = 381;

    bool public PUBLIC_SALE_STARTED;
    uint8 public MAX_PUBLIC_SALE_MINTS_PER_WALLET = 3;
    uint16 public MAX_TYPE1_MINTS = 2700;
    uint16 public MAX_TYPE2_MINTS = 300;

    address public projectWallet;

    /** PRIVATE VARS */
    mapping(address => bool) private _admins;
    mapping(address => uint256) private _type2AllowList;
    mapping(address => uint256) private _type2Mints;

    mapping(address => uint256) private _preSaleAddresses;
    mapping(address => uint256) private _preSaleMints;

    mapping(address => uint256) private _publicSaleMints;
    
    /** MODIFIERS */
    modifier onlyAdmin() {
        require(_admins[_msgSender()], "Main: Only admins can call this");
        _;
    }

    modifier onlyEOA() {
        require(tx.origin == _msgSender(), "Main: Only EOA");
        _;
    }

    modifier requireVariablesSet() {
        require(address(nftContract) != address(0), "Main: Nft contract not set");
        require(address(projectWallet) != address(0), "Main: Project wallet address must be set");
        _;
    }

    /** PUBLIC FUNCTIONS */
    function mint(uint256 amount, uint8 tokenType) external payable whenNotPaused nonReentrant onlyEOA {
        require(amount > 0, "Main: Amount must be greater than 0");
        require(PRE_SALE_STARTED || PUBLIC_SALE_STARTED, "Main: Sale has not started yet");
        if (PRE_SALE_STARTED) {
            require(_preSaleAddresses[_msgSender()] > 0, "Main: You are not on the whitelist");
            require(_preSaleMints[_msgSender()] + amount <= _preSaleAddresses[_msgSender()], "Main: You cannot mint more during pre-sale");
            require(nftContract.totalMinted() + amount <= MAX_PRE_SALE_MINTS, "Main: All pre-sale NFTs minted");
        } else {
            require(_publicSaleMints[_msgSender()] + amount <= MAX_PUBLIC_SALE_MINTS_PER_WALLET, "Main: You cannot mint more NFTs");
        }
        require(tokenType == 1, "Main: Token type must be 1");
        require(totalType1Minted + amount <= MAX_TYPE1_MINTS, "Main: Cannot exceed max-mint for Popsteak NFTs");
        require(mintPriceType1 > 0, "Main: Mint price cannot be 0");
        require(msg.value >= amount * mintPriceType1, "Main: Invalid payment amount");
        
        uint16[] memory tokenIds = new uint16[](amount);

        for (uint i = 0; i < amount; i++) {
            if (PRE_SALE_STARTED) {
                _preSaleMints[_msgSender()]++;
            } else {
                _publicSaleMints[_msgSender()]++;
            }

            nftContract.mint(_msgSender(), tokenType);
            tokenIds[i] = nftContract.totalMinted();
        }

        totalType1Minted += amount;
        emit ManyNftsMinted(_msgSender(), tokenIds);
    }

    function mintPopsteakX(address to, uint256 amount) external payable whenNotPaused nonReentrant onlyEOA {
        require(to != address(0), "Main: To cannot be the null address");
        require(amount > 0, "Main: Amount must be greater than 0");
        require(_type2AllowList[_msgSender()] > 0, "Main: Not allowed to mint Popsteak-X NFTs");
        require(totalType2Minted + amount <= MAX_TYPE2_MINTS, "Main: Cannot exceed max-mint for Popsteak-X NFTs");
        require(_type2Mints[_msgSender()] + amount <= _type2AllowList[_msgSender()], "Main: You cannot mint more Popsteak-X NFTs");
        require(msg.value >= amount * mintPriceType2, "Main: Invalid payment amount");

        uint16[] memory tokenIds = new uint16[](amount);

        for (uint i = 0; i < amount; i++) {
            // mint tokenType = 2 to the desired address
            nftContract.mint(to, 2);
            tokenIds[i] = nftContract.totalMinted();
        }
        
        // increase counters
        _type2Mints[_msgSender()] += amount;
        totalType2Minted += amount;
        emit ManyNftsMinted(to, tokenIds);
    }

    function getRemainingPreSaleMints(address addr) public view returns (uint256) {
        if (_preSaleAddresses[addr] >= _preSaleMints[addr]) {
            return _preSaleAddresses[addr] - _preSaleMints[addr];
        } 
        return 0;
    }

    function getRemainingPublicSaleMints(address addr) public view returns (uint256) {
        if (MAX_PUBLIC_SALE_MINTS_PER_WALLET >= _publicSaleMints[addr]) {
            return MAX_PUBLIC_SALE_MINTS_PER_WALLET - _publicSaleMints[addr];
        } 
        return 0;
    }

    function getRemainingType2Mints(address addr) public view returns (uint256) {
        if (_type2AllowList[addr] >= _type2Mints[addr]) {
            return _type2AllowList[addr] - _type2Mints[addr];
        } 
        return 0;
    }

    function isAdmin(address addr) external view returns(bool) {
        return (_admins[addr]);
    }

    /** ADMIN ONLY FUNCTIONS */
    // presale is for type1 tokens only
    function addToPresale(address[] memory addresses, uint256 allowedToMint) external onlyAdmin {
         for (uint i = 0; i < addresses.length; i++) {
            _preSaleAddresses[addresses[i]] = allowedToMint;
         }
    }

    // type2 tokens can only be minted by a few addresses as those are sold to millionaires and billionaires
    function addToType2AllowList(address[] memory addresses, uint16 allowedToMint) external onlyAdmin {
         for (uint i = 0; i < addresses.length; i++) {
            _type2AllowList[addresses[i]] = allowedToMint;
         }
    }

    function setMintPriceType1(uint256 number) external onlyAdmin {
        mintPriceType1 = number;
    }

    function setMintPriceType2(uint256 number) external onlyAdmin {
        mintPriceType2 = number;
    }

    function setTotalType1Minted(uint256 number) external onlyAdmin {
        totalType1Minted = number;
    }

    function setTotalType2Minted(uint256 number) external onlyAdmin {
        totalType2Minted = number;
    }

    function setPreSaleStarted(bool started) external onlyAdmin {
        PRE_SALE_STARTED = started;
        PUBLIC_SALE_STARTED = !PRE_SALE_STARTED;
    }

    function setMaxPreSaleMints(uint8 number) external onlyAdmin {
        MAX_PRE_SALE_MINTS = number;
    }

    function setPublicSaleStarted(bool started) external onlyAdmin {
        PUBLIC_SALE_STARTED = started;
        PRE_SALE_STARTED = !PUBLIC_SALE_STARTED;
    }

    function setMaxPublicSaleMintsPerWallet(uint8 number) external onlyAdmin {
        MAX_PUBLIC_SALE_MINTS_PER_WALLET = number;
    }

    /** OWNER ONLY FUNCTIONS */
    function setContracts(address _nftContract) external onlyOwner {
        nftContract = iNft(_nftContract);
    }

    function setPaused(bool _paused) external requireVariablesSet onlyOwner {
        if (_paused) _pause();
        else _unpause();
    }

    // team can mint only type1 tokens
    function mintForTeam(address receiver, uint256 amount) external whenNotPaused onlyOwner {
        require(totalType1Minted + amount <= MAX_TYPE1_MINTS, "Main: Cannot exceed max-mint for Popsteak NFTs");

        for (uint i = 0; i < amount; i++) {
            // mint token type 1
            nftContract.mint(receiver, 1);
        }
        
        totalType1Minted += amount;
    }

    function withdraw() external onlyOwner {
        uint256 totalAmount = address(this).balance;

        bool sent;
        (sent, ) = projectWallet.call{value: totalAmount}("");
        require(sent, "Main: Failed to send funds to projectWallet");
    }

    function setProjectWallet(address addr) external onlyOwner {
        projectWallet = addr;
    }

    function addAdmin(address addr) external onlyOwner {
        _admins[addr] = true;
    }

    function removeAdmin(address addr) external onlyOwner {
        delete _admins[addr];
    }
}
// File: /contracts/interfaces/iNft.sol



// SPDX-License-Identifier: MIT LICENSE

pragma solidity ^0.8.17;

import "@openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol";

interface iNft is IERC721Enumerable {

    // store lock meta data
    struct Trait {
        uint256 tokenId;
        uint8 tokenType; // 1=Popsteak, 2=Popsteak-X
    }

    function maxTokens() external returns (uint256);
    function totalMinted() external returns (uint16);

    function getWalletOfOwner(address owner) external view returns (uint256[] memory);
    function getLicenseURI() external view returns (string memory);

    function burn(uint256 tokenId) external;

    function mint(address recipient, uint8 tokenType) external; // onlyAdmin
    function setLicenseURI(string memory uri) external; // onlyOwner
}
// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/extensions/IERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}